

Name:	Lehrgang: Betriebssysteme	Datum:
Arbeitsblatt Nr.	Windows Scripting Host: Das File System Object (FSO)	Seite 1 von 2

## Arbeiten mit Text-Dateien

In VBScript lassen sich Text-Dateien sehr einfach erstellen und bearbeiten. Grundlage ist die Erzeugung eines sogenannten **TextStream**-Objektes. Dieses kann durch das **FileSystemObject** generiert werden. Am Besten zeige ich Ihnen das mal an einem Code-Schnipsel.

```
'VBText01.vbs
'erzeugt eine Textdatei, schreibt Text hinein

Option Explicit

'FileSystemObject erzeugen
Dim fso
Set fso = CreateObject("Scripting.FileSystemObject")

'TextStream-Object erzeugen
Dim datei
Set datei = fso.CreateTextFile("D:\MyText.txt")

'Jetzt kann man in die (noch leere) Datei Text schreiben
'Die Methode WriteLine schreibt einen Text in die geöffnete
'Datei und fügt einen Zeilenumbruch (Carriage Return/Linefeed)
'ans Ende der Zeile. Ohne Text wird eine Leerzeile erzeugt

datei.WriteLine "Das ist jetzt keine leere Datei mehr!"
datei.WriteLine "Ein Kreter sagt: Alle Kreter lügen!"
datei.WriteLine "Löst jemand dieses Paradoxon auf?"

'Schreibe eine Leerzeile
datei.WriteLine

datei.WriteLine "Jetzt ist aber Schluss!"

'Datei wird geschlossen
datei.close
```

Sie sehen: Vier Schritte sind erforderlich.

1. Erzeugen eines FileSystemObjects (falls noch keines zuvor für andere Zwecke erzeugt wurde)
2. Erzeugen des TextStream-Objektes mit CreateTextFile()
3. Schreiben in die nun geöffnete Datei mit WriteLine
4. Schließen der Datei mit Close, wenn nichts mehr zu schreiben ist.

### Übungen:

1. Informieren Sie sich in der Scripting-Hilfe-Datei über das TextStream-Objekt und die Methoden WriteLine bzw. Write. Wodurch unterscheiden sich die beiden Methoden?
2. Erstellen Sie das obige VBScript **VBText01.vbs** und bringen Sie es zur Ausführung. Sehen Sie sich das Ergebnis mit dem Notepad an. Ersetzen Sie die WriteLine-Methode durch die Write-Methode. Sehen Sie sich nun dieses Ergebnis mit dem Notepad an. Was ist eigentlich mit der ursprünglichen Datei (mit WriteLine erstellt) passiert?
3. Erstellen Sie ein VBScript namens **VBCreateScriptFile.vbs**, welches von ihnen per InputBox einen Dateinamen verlangt (ohne **.vbs**! Das soll im Skript angehängt werden!) und anschließend diese Datei erstellt. In die Datei soll nebenstehendes eingefügt werden. Wird kein Name eingegeben oder wird die InputBox mit "Abbrechen" beendet, soll keine Datei erstellt werden.

```
'*****
'eingegabener Name der Datei
'*****
Option Explicit
```

Name:	Lehrgang: Betriebssysteme	Datum:
Arbeitsblatt Nr.	Windows Scripting Host: Das File System Object (FSO)	Seite 2 von 2

## Lesen aus Text-Dateien

Das Lesen aus Text-Dateien ist ähnlich einfach wie das Schreiben; naja vielleicht *etwas* schwieriger. Da nun die Datei ja bereits existiert, kommt hier die Methode **OpenTextFile** zum Zuge. Auch hier ein Code-Schnipsel, der das Lesen von Text-Dateien verdeutlichen soll.

```
'VBText02.vbs
'liest eine Textdatei und gibt den Inhalt mit der MsgBox aus

Option Explicit

'FileSystemObject erzeugen
Dim fso
Set fso = CreateObject("Scripting.FileSystemObject")

'Für das Beispiel sollte die Textdatei existieren!
If fs.FileExists("D:\MyText.txt") then

    'TextStream-Object erzeugen
    Dim datei
    Set datei = fso.OpenTextFile("D:\MyText.txt")

    'Variable für den Dateiinhalt
    Dim Inhalt

    'Jetzt wird die Datei zeilenweise gelesen; und zwar bis zum Dateiende
    'Am Schluß erfolgt die Ausgabe mit der MsgBox
    'AtEndOfStream ist eine Eigenschaft, die angibt, ob noch Zeichen zu lesen sind
    do until datei.AtEndOfStream
        Inhalt = datei.ReadLine
    loop
    datei.close
    MsgBox Inhalt,, "Inhalt von D:\MyText.txt"

else
    MsgBox "Zu schnell gelöscht!?! Datei gibt es nicht!",vbInformation,"Fehler"
end if
```

Also, war **doch** gar nicht soo schwer :^)!

### Übungen:

1. Informieren Sie sich in der Hilfe-Datei über die weiteren Methoden **Read** und **ReadAll**.

**Read** liest \_\_\_\_\_

**ReadAll** liest \_\_\_\_\_

- Erstellen Sie das obige VBScript **VBText02.vbs** und bringen Sie es zur Ausführung. In der Datei ist bei genauem Abtippen ein kleiner Fehler, der zu einem Runtime-Error führt! Wenn dieser beseitigt ist, überlegen Sie, warum die Ausgabe nur eine Zeile beträgt. Was muss geändert werden? (Tipp: Innerhalb der Schleife) sollten Sie den Fehler gefunden haben, könnte noch der Fehler vorhanden sein, dass nun der Dateiinhalt als eine lange Zeile angezeigt wird. Was ist nun noch zu ergänzen?
- Informieren Sie sich in der Hilfe, wie an eine Datei Text angefügt werden kann (Append!).
- Erstellen Sie ein VBScript namens **VBText03.vbs**, das an die obige Datei zwei beliebige Textzeilen anfügt.